

# MAPLE MANUAL FOR THE ETA PACKAGE

FRANK GARVAN

**Abstract.** This is a tutorial for using ETA, a MAPLE package for calculating with Dedekind's eta function. The ETA package is designed for proving eta-product identities using the valence formula for modular functions.

## 1. Introduction

The Dedekind eta-function is defined by

$$\eta(\tau) = q^{\frac{1}{24}} \prod_{n=1}^{\infty} (1 - q^n)$$

where  $\tau \in \mathcal{H} = \{z \in \mathbb{C} : \Im(z) > 0\}$  and  $q = e^{2\pi i \tau}$ .

The main goal of the ETA maple package is to automatically prove identities for eta-products.

### 1.1 Installation Instructions

First install the qseries package from

<http://qseries.org/fgarvan/qmaple/qseries>

and follow the directions on that page. Before proceeding it is advisable to become familiar with the functions in the qseries package. See [3] for a tutorial. Then go to

<http://qseries.org/fgarvan/qmaple/ETA>

to install the ETA package.

## 2. Modularity of eta-products

To prove a given eta-function identity one needs to basically do the following.

- Rewrite the identity in terms of generalized eta-functions.
- Check that each term in the identity is a modular function on some group  $\Gamma_0(N)$ .
- Determine the order at each cusp of  $\Gamma_0(N)$  of each term in the identity.
- Use the valence formula to determine up to which power of  $q$  is needed to verify the identity.
- Finally prove the identity by carrying out the verification.

In this section we explain how to carry out each of these steps in maple. Then we show how the whole process of proof can be automated.

### 2.1. Encoding eta-functions.

We encode  $\eta(\tau)$  by **eta (tau)**. We will consider eta-products of the form

$$f(\tau) = \prod_{d|N} \eta(d\tau)^{m_d}$$

where  $N$  is a positive integer,  $d > 0$  and  $m_d \in \mathbb{Z}$ . We encode the product

$$\prod_{j=1}^m \eta(t_j \tau)^{r_j} \quad (2.2)$$

by the list

$$[t_1, r_1, t_2, r_2, \dots, t_m, r_m]$$

We call such a product an eta-product. Such a list is called a *generalized permutation* and is usually written symbolically as

$$t_1^{r_1} t_2^{r_2} \dots t_m^{r_m}$$

We use the functions **GPmake** and **gp2etaproduct** to convert between symbolic forms of eta-products.

**GPmake (etaproduct)** --- converts an eta-product to a generalized permutation.

**gp2etaproduct (gp)** --- converts the generalized permutation **gp** to an eta-product.

EXAMPLE:

```
> with(qseries) :
> with(ETA) :
> HM5:=q*mul((1-q^(5*n))^6/(1-q^n)^6, n=1..100) :
> ep:=etaproduct(HM5, q, 50) ;
```

$$ep := \frac{\eta(5\tau)^6}{\eta(\tau)^6} \quad (1)$$

```
> gp:=GPmake(ep) ;
```

$$gp := [5, 6, 1, -6] \quad (2)$$

```
> gp2etaproduct(gp) ;
```

$$\frac{\eta(5\tau)^6}{\eta(\tau)^6} \quad (3)$$

## 2.2. q-Expansions.

There are two functions to compute q-expansions of eta-products.

**etaproducttoqseries (etaproduct)** --- returns the q-expansion of the eta-product **etaproduct** up to  $q^T$ .

**etaproducttoqseries2 (etaproduct)** --- a version of **etaproducttoqseries** that omits the factor  $q^{\frac{1}{24}}$

in the q-expansion of  $\eta(\tau)$ .

EXAMPLE:

```
> with(qseries) :
> with(ETA) :
> gp:=[2, 2, 1, -1] :
> ep:=gp2etaproduct(gp) ;
```

$$ep := \frac{\eta(2\tau)^2}{\eta(\tau)} \quad (4)$$

```
> etaproducttoqseries(ep, 50) ;
```

$$q^{1/8} + q^{9/8} + q^{25/8} + q^{49/8} + q^{81/8} + q^{\frac{121}{8}} + q^{\frac{169}{8}} + q^{\frac{225}{8}} + q^{\frac{289}{8}} + q^{\frac{361}{8}} + O\left(q^{\frac{401}{8}}\right) \quad (5)$$

```
> etaproducttoqseries2(ep, 50) ;
```

$$1 + q + q^3 + q^6 + q^{10} + q^{15} + q^{21} + q^{28} + q^{36} + q^{45} + O(q^{50}) \quad (6)$$

### 2.3. Checking modularity.

Newman [6] has found necessary and sufficient conditions under which an eta-product is a modular function on  $\Gamma_0(N)$ . Let  $N > 0$  be a fixed integer. Here an eta-product takes the form

$$f(\tau) = \prod_{d|N} \eta(d\tau)^{m_d}$$

where each  $d > 0$  and  $m_d \in \mathbb{Z}$ .

Theorem 2.1 (Theorem 4.7, [6]). The function  $f(\tau)$  (given above) is a modular function on  $\Gamma_0(N)$  if and only if

1.  $\sum_{d|N} m_d = 0$ .
2.  $\sum_{d|N} d m_d \equiv 0 \pmod{24}$
3.  $\sum_{d|N} \frac{N}{d} m_d \equiv 0 \pmod{24}$
4.  $\prod_{d|N} d^{|m_d|}$  is a square.

By this theorem the eta-product (2.2) corresponding the generalized permutation

$$[t_1, r_1, t_2, r_2, \dots, t_m, r_m]$$

is

a modular function on  $\Gamma_0(N)$  if

1.  $\sum_{j=1}^m r_j = 0$ .
2.  $\sum_{j=1}^m t_j r_j \equiv 0 \pmod{24}$
3. The integer  $\prod_{j=1}^m t_j^{|r_j|}$  is a square.
4. For each  $j$ ,  $r_j \neq 0$  and  $t_j | N$ .
5.  $\sum_{j=1}^m \frac{N}{t_j} r_j \equiv 0 \pmod{24}$ .

**gammacheck(gp, N)** --- checks whether the eta-product corresponding to the generalized permutation **gp** is a modular function on  $\Gamma_0(N)$ . To see a test of each condition set the following global variable `xprint := true`.

```
> with(qseries) :
> with(ETA) :
> gp := [1, 2, 2, -1, 10, 1, 5, -2] ;
      gp := [1, 2, 2, -1, 10, 1, 5, -2]
```

```
> ep:=gp2etaprod(gp);
```

$$ep := \frac{\eta(\tau)^2 \eta(10\tau)}{\eta(2\tau) \eta(5\tau)^2} \quad (8)$$

```
> gammacheck(gp,10);
```

$$0 \quad (9)$$

```
> ep2:=ep^2;
```

$$ep2 := \frac{\eta(\tau)^4 \eta(10\tau)^2}{\eta(2\tau)^2 \eta(5\tau)^4} \quad (10)$$

```
> gp2:=GPmake(ep2);
```

$$gp2 := [1, 4, 2, -2, 10, 2, 5, -4] \quad (11)$$

```
> gammacheck(gp2,10);
```

$$1 \quad (12)$$

We considered two etaproduts:

$$g_1(\tau) = \frac{\eta(\tau)^2 \eta(10\tau)}{\eta(2\tau) \eta(5\tau)^2}, \quad g_2(\tau) = \frac{\eta(\tau)^4 \eta(10\tau)^2}{\eta(2\tau)^2 \eta(5\tau)^4}.$$

We see that  $g_1(\tau)$  is not a modular function on  $\Gamma_0(10)$  but its square  $g_2(\tau)$  is a modular function on  $\Gamma_0(10)$ . To see the reasons why  $g_1(\tau)$  failed we set the global variable `xprint` to `true`.

```
> xprint:=true;
```

```
> gammacheck(gp,10);
```

*Condition (1) holds*  
*Condition (2) holds*  
*Condition (3) does not hold*  
*Condition (4) holds*  
*Condition (5) does not hold*  
*function is NOT invariant*

$$0 \quad (13)$$

We see that  $g_1(\tau)$  is not a modular function on  $\Gamma_0(10)$  since it failed to satisfy Conditions (3) and (5).

To check whether an eta-product is a modular form with character use the following function.

**gamma0FORMCHECK(gp,N)** --- checks whether the eta-product corresponding to the generalized permutation **gp** is a form on  $\Gamma_0(N)$  with character. See [7].

```
> gp2:=[1,4,2,4,4,-3,10,2,20,-1];
```

```
> ep2:=gp2etaprod(gp2);
```

$$ep2 := \frac{\eta(\tau)^4 \eta(2\tau)^4 \eta(10\tau)^2}{\eta(4\tau)^3 \eta(20\tau)} \quad (14)$$

```
> gamma0FORMCHECK(gp2,40);
```

$$["N=", 40, "weight=", 3, "character=", [-(2)^{14} (5)^3, d]] \quad (15)$$

This means that the function

$$g(\tau) = \frac{\eta(\tau)^4 \eta(2\tau)^4 \eta(10\tau)^2}{\eta(4\tau)^3 \eta(20\tau)}$$

is a modular form on  $\Gamma_0(40)$  of weight 3 and character  $\chi(d) = \left( \frac{-2048000}{d} \right) = \left( \frac{-20}{d} \right)$ .

## 2.4. Cusps.

Chua and Lang [2] have found a set of inequivalent cusps for  $\Gamma_0(N)$ .

**Theorem 2.2** ([2](p.354)). Let  $N$  be a positive integer and for each positive divisor  $d$  of  $N$  let

$$e_d = \left( d, \frac{N}{d} \right).$$

Then the set

$$\Delta = \bigcup_{d|N} S_d$$

is a complete set of inequivalent cusps of  $\Gamma_0(N)$  where

$$S_d = \left\{ \frac{x_i}{d} : (x_i, d) = 1, 0 \leq x_i \leq d-1, x_i \not\equiv x_j \pmod{e_d} \right\}$$

**cuspmake(N)** --- returns a complete set of inequivalent cusps of  $\Gamma_0(N)$ .

EXAMPLE

> **cuspmake(40);**

$$\left\{ 0, \frac{1}{2}, \frac{1}{4}, \frac{1}{5}, \frac{1}{8}, \frac{1}{10}, \frac{1}{20}, \frac{1}{40} \right\} \quad (16)$$

Biagioli [1] has found the fan width of the cusps of  $\Gamma_0(N)$ .

**Lemma 2.3** (Lemma 4.2, [1]). If  $(r, s) = 1$ , then the fan width of  $\Gamma_0(N)$  at  $\frac{r}{s}$  is

$$\kappa(\Gamma_0(N), \zeta) = \frac{N}{(N, s^2)}.$$

**fanwidth(r, N)** --- returns the width of the cusp  $\mathbf{r}$  for the group  $\Gamma_0(N)$ .

> **fanwidth(1/8, 40);**

$$5 \quad (17)$$

## 2.5. Orders at cusps.

Ligozat [5] has computed the order of an eta-product at the cusps of  $\Gamma_0(N)$ .

**Theorem 2.4** (Theorem 4.8, [5]). If the eta-product  $f(\tau)$  (given in (2.3))

is a modular function on  $\Gamma_0(N)$ , then its invariant order at the cusp  $s = \frac{b}{c}$  (assuming  $(b, c) = 1$ ) is

$$\text{ord}(f(\tau); s) = \sum_{d|N} \frac{(d, c)^2 m_d}{24 d}.$$

Following [1, p.275], [8, p.91] we consider the order of a function  $f$  with respect to a congruence subgroup  $\Gamma$  at the cusp  $\zeta \in \mathbb{Q} \cup \{\infty\}$  and denote this by

$$(2.6) \quad \text{ORD}(f, \zeta, \Gamma) = \kappa(\Gamma, \zeta) \text{ord}(f; \zeta)$$

The following are functions for computing orders and invariant orders of eta-products at cusps.

**cusppord(etaProd, cusp)** --- computes the invariant order at the given cusp of the given eta-product.  
**cusppord(etaProd, N, cusp)** --- computes the order of the given eta-product at the given cusp with respect to the group  $\Gamma_0(N)$ .  
**cusppords(etaProd, CUSPS, N)** --- computes the order of the given eta-product at each cusp in a set respect to the group  $\Gamma_0(N)$ .

EXAMPLE

> gp := [20, -3, 10, 5, 5, -2, 4, 15, 2, -25, 1, 10];  
 > ep := gp2etaProd(gp);

$$ep := \frac{\eta(10\tau)^5 \eta(4\tau)^{15} \eta(\tau)^{10}}{\eta(20\tau)^3 \eta(5\tau)^2 \eta(2\tau)^{25}} \quad (18)$$

> gammacheck(gp, 20);

*Condition (1) holds*  
*Condition (2) holds*  
*Condition (3) holds*  
*Condition (4) holds*  
*Condition (5) holds*  
*function is invariant*

$$1 \quad (19)$$

> cusppord(ep, 1/4);

$$\frac{4}{5} \quad (20)$$

> fanwidth(1/4, 20);

$$5 \quad (21)$$

> cusppord(ep, 20, 1/4);

$$4 \quad (22)$$

> cusppords(ep, cusppmake(20), 20);

$$\left[ [0, 1], \left[ \frac{1}{2}, -5 \right], \left[ \frac{1}{4}, 4 \right], \left[ \frac{1}{5}, 0 \right], \left[ \frac{1}{10}, 0 \right], \left[ \frac{1}{20}, 0 \right] \right] \quad (23)$$

Let

$$g(\tau) = \frac{\eta(10\tau)^5 \eta(4\tau)^{15} \eta(\tau)^{10}}{\eta(20\tau)^3 \eta(5\tau)^2 \eta(2\tau)^{25}}$$

We see that  $g(\tau)$  is a modular function on  $\Gamma_0(20)$ ,

$$\text{ord}\left(g(\tau); \frac{1}{4}\right) = \frac{4}{5}$$

and

$$\text{ORD}(g, 1/4, \Gamma_0(20)) = \kappa(\Gamma_0(20), \frac{1}{4}) \text{ord}(g; 1/4) = 5 \cdot \frac{4}{5} = 4.$$

### 3. Proving eta-function identities

#### 3.1. Linear relations between eta-products.

1. Rewrite the identity in terms of eta-functions.
2. Check that each term in the identity is a modular function on some group  $\Gamma_0(N)$ .
3. Determine the order at each cusp of  $\Gamma_0(N)$  of each term in the identity.
4. Use the valence formula to determine up to which power of  $q$  is needed to verify the identity.
5. Finally prove the identity by carrying out the verification.

In this section we explain how to carry out each of these steps in maple. Then we show how the whole process of proof can be automated.

Our method for proving eta-product identities depends on

**Theorem 3.1** (The Valence Formula [8](p.98)). Let  $f(\tau)$  be a modular form of weight  $k$  with respect to a subgroup  $\Gamma$  of finite index in  $\Gamma(1) = SL_2(\mathbb{Z})$ . Then

$$(3.1) \text{ORD}(f, \Gamma) = \sum_{\zeta \in R^*} \text{ORD}(f, \zeta, \Gamma) = \frac{1}{12} \mu k,$$

where  $\mu$  is the index of  $\hat{\Gamma}$  in  $\widehat{\Gamma(1)}$ ,  $R^*$  is a fundamental region for  $\Gamma$ , and  $\text{ORD}(f, \zeta, \Gamma)$  is given in equation (2.6).

*Remark 3.2*

For  $\zeta \in \mathcal{H}$ ,  $\text{ORD}(f, \zeta, \Gamma)$  is defined in terms of the invariant order  $\text{ord}(f; \zeta)$  which is interpreted in the usual sense. See [8, p.91] for details of this and the notation used.

Since any modular function has weight  $k = 0$  and any eta-product has no zeros and no poles on the upper-half plane we have

**Corollary 3.3.** Let  $f_1(\tau), f_2(\tau), \dots, f_n(\tau)$  be eta-products that are modular functions on  $\Gamma_0(N)$ . Let  $S_N$  be a set of inequivalent cusps for  $\Gamma_0(N)$ . Define the constant

$$(3.2) \quad B = \sum_{s \in S_N, s \neq \infty} \min ( \{ \text{ORD}(f_j, s, \Gamma_0(N)) : 1 \leq j \leq n \} \cup \{0\} ),$$

and consider

$$(3.3) \quad g(\tau) := \alpha_1 f_1(\tau) + \alpha_2 f_2(\tau) + \dots + \alpha_n f_n(\tau) + 1.$$

Then

$$g(\tau) \equiv 0$$

if and only if

$$(3.4) \quad \text{ORD}(g(\tau), \infty, \Gamma_0(N)) > -B.$$

To prove an alleged eta-product identity, we first rewrite it in the form

$$(3.5) \quad \alpha_1 f_1(\tau) + \alpha_2 f_2(\tau) + \dots + \alpha_n f_n(\tau) + 1 = 0,$$

where each  $\alpha_j \in \mathbb{C}$  and each  $f_j(\tau)$  is an eta-product of level  $N$ . We use the following algorithm:

**STEP 0.** Write the identity in the form (3.5).

**STEP 1.** Use Theorem 2.1 to check that  $f_j(\tau)$  is a modular function on  $\Gamma_0(N)$  for each  $1 \leq j \leq n$ .

**STEP 2.** Use Theorem 2.2 to find a set  $S_N$  of inequivalent cusps for  $\Gamma_0(N)$  and the fan width of each cusp.

**STEP 3.** Use Theorem 2.4 to calculate the order of each eta-product  $f_j(\tau)$  at each cusp of  $\Gamma_0(N)$ .

**STEP 4.** Calculate  $B = \sum_{s \in S_N, s \neq \infty} \min ( \{ ORD(f, s, \Gamma_0(N)) : 1 \leq j \leq n \} \cup \{0\} )$ .

**STEP 5.** Show that  $ORD(g(\tau), \infty, \Gamma_0(N)) > -B$ ,

where  $g(\tau) = \alpha_1 f_1(\tau) + \alpha_2 f_2(\tau) + \dots + \alpha_n f_n(\tau) + 1$ .

Corollary 3.3 then implies that  $g(\tau) \equiv 0$  and hence the eta-product identity (3.5).

To calculate the constant B we use

**mintotORDS(L, n)** --- returns the constant B in equation (3.2) where L is the array of ORDS:

$$L := [ ORD(f_1), ORD(f_1), \dots, ORD(f_n) ],$$

$$ORD(f) = [ ORD(f, \zeta_1, \Gamma_0(N)), ORD(f, \zeta_2, \Gamma_0(N)), \dots, ORD(f, \zeta_m, \Gamma_0(N)) ]$$

and  $\zeta_1, \zeta_2, \dots, \zeta_m$  are the inequivalent cusps of  $\Gamma_0(N)$  apart from  $\infty$ . Each  $ORD(f)$  is computed using **getaprodcuspORDS**.

**EXAMPLE:** As an example we prove Ramanujan's Entry 3.1:

$$(3.6) \quad PQ + \frac{9}{PQ} = \left( \frac{Q}{P} \right)^3 + \left( \frac{P}{Q} \right)^3,$$

where

$$P = \frac{\eta(\tau)^2}{\eta(3\tau)^2} \quad \text{and} \quad Q = \frac{\eta(2\tau)^2}{\eta(6\tau)^2}$$

> **with(qseries) :**

> **with(ETA) :**

> **xprint:=false:**

> **gpP:=[1,2,3,-2] :**

> **gpQ:=[2,2,6,-2] :**

> **P:=gp2etaproduct(gpP) ;**

$$P := \frac{\eta(\tau)^2}{\eta(3\tau)^2} \tag{24}$$

> **Q:=gp2etaproduct(gpQ) ;**

$$Q := \frac{\eta(2\tau)^2}{\eta(6\tau)^2} \tag{25}$$

> **ETAid:=P\*Q+9/P/Q - (Q/P)^3 - (P/Q)^3;**

$$ETAid := \frac{\eta(\tau)^2 \eta(2\tau)^2}{\eta(3\tau)^2 \eta(6\tau)^2} + \frac{9 \eta(3\tau)^2 \eta(6\tau)^2}{\eta(\tau)^2 \eta(2\tau)^2} - \frac{\eta(2\tau)^6 \eta(3\tau)^6}{\eta(6\tau)^6 \eta(\tau)^6} - \frac{\eta(\tau)^6 \eta(6\tau)^6}{\eta(3\tau)^6 \eta(2\tau)^6} \tag{26}$$

> **ETAidn:=etanormalid(%);**

$$ETAidn := 1 + \frac{9 \eta(3\tau)^4 \eta(6\tau)^4}{\eta(\tau)^4 \eta(2\tau)^4} - \frac{\eta(3\tau)^8 \eta(2\tau)^4}{\eta(\tau)^8 \eta(6\tau)^4} - \frac{\eta(\tau)^4 \eta(6\tau)^8}{\eta(3\tau)^4 \eta(2\tau)^8} \tag{27}$$

Thus identity (3.6) is equivalent to

$$g(\tau) = 0,$$

where

$$g(\tau) = 1 + 9 f_1(\tau) - f_2(\tau) - f_3(\tau),$$



$$f_1(\tau) = \frac{\eta(3\tau)^4 \eta(6\tau)^4}{\eta(\tau)^4 \eta(2\tau)^4}, \quad f_2(\tau) = \frac{\eta(3\tau)^8 \eta(2\tau)^4}{\eta(\tau)^8 \eta(6\tau)^4}, \quad f_3(\tau) = \frac{\eta(\tau)^4 \eta(6\tau)^8}{\eta(3\tau)^4 \eta(2\tau)^8}.$$

STEP 1. We check that each eta-product is a modular function on  $\Gamma_0(6)$ .

**> f1:=op(2,ETAidn)/9;**

$$f1 := \frac{\eta(3\tau)^4 \eta(6\tau)^4}{\eta(\tau)^4 \eta(2\tau)^4} \quad (28)$$

**> f2:=-op(3,ETAidn);**

$$f2 := \frac{\eta(3\tau)^8 \eta(2\tau)^4}{\eta(\tau)^8 \eta(6\tau)^4} \quad (29)$$

**> f3:=-op(4,ETAidn);**

$$f3 := \frac{\eta(\tau)^4 \eta(6\tau)^8}{\eta(3\tau)^4 \eta(2\tau)^8} \quad (30)$$

**> gpf1:=GPmake(f1): gpf2:=GPmake(f2): gpf3:=GPmake(f3):**

**> gammacheck(gpf1,6), gammacheck(gpf2,6), gammacheck(gpf3,6);**  
1, 1, 1 (31)

STEP 2. We find a set of inequivalent cusps for  $\Gamma_0(6)$  and their fan widths.

**> C6:=cuspmake(6);**

$$C6 := \left\{ 0, \frac{1}{2}, \frac{1}{3}, \frac{1}{6} \right\} \quad (32)$$

**> seq([cusp, fanwidth(cusp,6)], cusp in C6);**

$$[0, 6], \left[ \frac{1}{2}, 3 \right], \left[ \frac{1}{3}, 2 \right], \left[ \frac{1}{6}, 1 \right] \quad (33)$$

STEP 3. We compute  $ORD(f_j, \zeta, \Gamma_0(6))$  for each  $j$  and each cusp  $\zeta$  of  $\Gamma_0(6)$  apart from  $\infty$ .

**> C6:=cuspmake(6) minus {1/6};**

$$C6 := \left\{ 0, \frac{1}{2}, \frac{1}{3} \right\} \quad (34)$$

**> ORDS0:=cuspORDSnotoo(1,C6,6);**

$$ORDS0 := \left[ [0, 0], \left[ \frac{1}{2}, 0 \right], \left[ \frac{1}{3}, 0 \right] \right] \quad (35)$$

**> ORDS1:=cuspORDSnotoo(f1,C6,6);**

$$ORDS1 := \left[ [0, -1], \left[ \frac{1}{2}, -1 \right], \left[ \frac{1}{3}, 1 \right] \right] \quad (36)$$

**> ORDS2:=cuspORDSnotoo(f2,C6,6);**

$$ORDS2 := \left[ [0, -1], \left[ \frac{1}{2}, 0 \right], \left[ \frac{1}{3}, 1 \right] \right] \quad (37)$$

**> ORDS3:=cuspORDSnotoo(f3,C6,6);**

$$ORDS3 := \left[ [0, 0], \left[ \frac{1}{2}, -1 \right], \left[ \frac{1}{3}, 0 \right] \right] \quad (38)$$

STEP 4. We calculate the constant B in (3.2).

**> mintotGAMMAORDS([ORDS0,ORDS1,ORDS2,ORDS3],4);**

$$-2 \quad (39)$$

STEP 5. To prove the identity (3.6) we need to verify that

$$ORD(g(\tau), \infty, \Gamma_0(6)) > 2$$

```
> qetacombo(1+9*f1-f2-f3,100);
```

0

(40)

This completes the proof of the identity (3.6). We only had to show that the coefficient of  $q^j$  was zero in the  $q$ -expansion of  $g(\tau)$  for  $j \leq 3$ . We actually did it for  $j \leq 100$  as a check.

STEPS 1-5 may be automated using the following function.

**provemodfuncGAMMA0id(etaid,N)** --- returns the constant  $B$  in equation (3.2) and prints details of the verification and proof of the identity corresponding to **etaid**, which is a linear combination of symbolic eta-products, and **N** is the level. If **xprint=true** then more details of the verification are printed. When this function is called there is a query asking whether to verify the identity. Enter *yes* to carry out the verification.

```
> provemodfuncGAMMA0id(1+9*f1-f2-f3,6);
```

```
"TERM ", 1, "of ", 4, " *****"
"TERM ", 2, "of ", 4, " *****"
"TERM ", 3, "of ", 4, " *****"
"TERM ", 4, "of ", 4, " *****"
"mintotord = ", -2
"TO PROVE the identity we need to show that v[oo](ID) > ", 2
*** There were NO errors.
*** o Each term was modular function on
    Gamma0(6).
*** o We also checked that the total order of
    each term was zero.
"*** WARNING: some terms were constants. ***"
"See array CONTERMS."
To prove the identity we will need to verify if up to
q^(3).
Do you want to prove the identity? (yes/no)
You entered yes.
We verify the identity to O(q^(14)).
RESULT: The identity holds to O(q^(14)).
CONCLUSION: This proves the identity since we had only
to show that v[oo](ID) > 2.
```

**printETAIDORDStable** --- prints an ORDs table for the  $f_j$  and lower bound for  $g$  after **provemodfuncGAMMA0id** is run. Formatted output from our example is given below in Table 3.1. By summing the last column we see that  $B = -2$ , which confirms an earlier calculation using **mintotORDS**.

```
> printETAIDORDStable();
```

```
-----
printETAIDORDStable()
  Print a table of ORDS for each term in a eta-prod-identity
  using global data produced by the function
  provemodfuncGAMMA0id.
  Table is stored in the matrix bigmat which is returned.
-----
ORDS Table for the jacprod identity
_G =
      _F1 + 9 _F2 - _F3 - _F4 = 0
```

where

$$\_F[0] =$$

$$1$$

$$\_F[1] =$$

$$\frac{\eta(3\tau)^4 \eta(6\tau)^4}{\eta(\tau)^4 \eta(2\tau)^4}$$

$$\_F[2] =$$

$$\frac{\eta(3\tau)^8 \eta(2\tau)^4}{\eta(\tau)^8 \eta(6\tau)^4}$$

$$\_F[3] =$$

$$\frac{\eta(\tau)^4 \eta(6\tau)^8}{\eta(3\tau)^4 \eta(2\tau)^8}$$

<i>cusp</i>	$ORD(\_F_0)$	$ORD(\_F_1)$	$ORD(\_F_2)$	$ORD(\_F_3)$	$ORD(\_G)$
0	0	-1	-1	0	-1
$\frac{1}{2}$	0	-1	0	-1	-1
$\frac{1}{3}$	0	1	1	0	0

The last column of the table gives a lower bound for ORDS of  $\_G$ . By summing this last column we see that the identity can be proved by showing that the coefficients  $q^0, q^1, \dots, q^3$  are all zero.

This confirms the calculation done by `provemodfuncGAMMA0id`.

*bigmat*

(41)

`provemodfuncGAMMA0idBATCH(etaid,N)` | is a version of `provemodfuncGAMMA0id` that prints less detail and does not query.

**> provemodfuncGAMMA0idBATCH(1+9\*f1-f2-f3,6);**

\*\*\* There were NO errors.

\*\*\* o Each term was modular function on  $\Gamma_0(6)$ .

\*\*\* o We also checked that the total order of each term was zero.

To prove the identity we will need to verify if up to  $q^3$ .

\*\*\* The identity below is PROVED!

[1, -2, 0]

(42)

Let  $L = [1, -2, 0]$ .  $L[1] = 1$  means the identity is proved.  $L[2] = B$  (the constant in equation (3.2)), and we see that  $B = -2$ . This confirms an earlier calculation using `mintotORDS`. To print out minimal information set `noprnt:=true`.

**> noprnt:=true:**

**> provemodfuncGAMMA0idBATCH(1+9\*f1-f2-f3,6);**

[1, -2, 0]

(43)

### 3.2. $U_p$ identities.

Let  $p$  be prime and suppose

$$f(\tau) = \sum_{n \geq n_0} a(n) q^n$$

is a modular function. We define

$$(3.8) \quad U_p f(\tau) = \sum_{pn \geq n_0} a(pn) q^n = \frac{1}{p} \sum_{k=0}^{p-1} f\left(\frac{\tau+k}{p}\right).$$

In this section show how to modify our method to prove eta-product identities for  $U_p f(\tau)$  when  $f(\tau)$  is also an eta-product.

It is known that if  $f(\tau)$  is a modular function on  $\Gamma_0(pN)$ , where  $p \mid N$ , then  $U_p f(\tau)$  is a modular function on  $\Gamma_0(N)$ . Gordon and Hughes [4, Theorem 4, p.336] have found lower bounds for the invariant orders of  $U_p f(\tau)$  at cusps. Let  $v_p(n)$  denote the  $p$ -adic order of an integer  $n$ ; i.e. the highest power of  $p$  that divides  $n$ .

**Theorem 3.4** (Theorem 4, [4]). Suppose  $f(\tau)$  is a modular function on  $\Gamma_0(pN)$ , where  $p$  is prime and  $p \mid N$ . Let  $r = \frac{\beta}{\delta}$  be a cusp of  $\Gamma_0(N)$ , where  $\delta \mid N$  and  $(\beta, \delta) = 1$ . Then

$$\text{ORD}(U_p f, r, \Gamma_0(N)) \geq \begin{cases} \frac{1}{p} \text{ORD}\left(f, \frac{r}{p}, \Gamma_0(pN)\right) & \text{if } v_p(\delta) \geq \frac{1}{2} v_p(N) \\ \text{ORD}\left(f, \frac{r}{p}, \Gamma_0(pN)\right) & \text{if } 0 < v_p(\delta) < \frac{1}{2} v_p(N) \\ \text{ORD}\left(f, \frac{r+k}{p}, \Gamma_0(pN)\right) & \text{if } v_p(\delta) = 0. \end{cases} \min_{0 \leq k \leq p-1} \text{ORD}\left(f, \frac{r+k}{p}, \Gamma_0(pN)\right)$$

```
> gpF := [2, 1, 25, 1, 1, -1, 50, -1] :
```

```
> epF := gp2etaprod(gpF) ;
```

$$epF := \frac{\eta(2\tau) \eta(25\tau)}{\eta(\tau) \eta(50\tau)} \quad (44)$$

```
> gammacheck(gpF, 50) ;
```

$$1 \quad (45)$$

```
> epF := gp2etaprod(gpF) ;
```

$$epF := \frac{\eta(2\tau) \eta(25\tau)}{\eta(\tau) \eta(50\tau)} \quad (46)$$

```
> seq([cusp, UpLB(epF, cusp, 50, 5)], cusp in cuspmake(10)) ;
```

$$\left[0, -1\right], \left[\frac{1}{2}, 0\right], \left[\frac{1}{5}, \frac{1}{5}\right], \left[\frac{1}{10}, -\frac{1}{5}\right] \quad (47)$$

We see that

$$F(\tau) = \frac{\eta(2\tau) \eta(25\tau)}{\eta(\tau) \eta(50\tau)}$$

is a modular function on  $\Gamma_0(50)$ . From Theorem 3.4 we have the following lower bounds for the orders of  $U_5F$  at the cusps of  $\Gamma_0(10)$ . It turns out that  $U_5F$  is an eta-product.

```
> F:=etaproductseries(epF,1000);
> sf:=sift(F,q,5,0,1000);
> epG:=etamake(sf,q,100);
```

$$epG := \frac{\eta(5\tau)^4 \eta(2\tau)^2}{\eta(10\tau)^2 \eta(\tau)^4} \quad (48)$$

```
> seq([cusp, cuspORD(epG,10,cusp)], cusp in cuspmake(10));
```

$$\left[0, -1\right], \left[\frac{1}{2}, 0\right], \left[\frac{1}{5}, 1\right], \left[\frac{1}{10}, 0\right] \quad (49)$$

We see that

$$U_5(F) = G,$$

where

$$G(\tau) = \frac{\eta(5\tau)^4 \eta(2\tau)^2}{\eta(10\tau)^2 \eta(\tau)^4}.$$

We have the following exact values for the orders of  $G(\tau)$  at the cusps of  $\Gamma_0(10)$ .

$\zeta$	0	1/2	1/5	1/10
$ORD(G, \zeta, \Gamma_0(10)) =$	-1	0	1	0

These values are consistent with the lower bounds that we found.

It is a simple matter to modify our method, from Section 3.1, for proving linear relations between eta-products, to proving  $U_p$  eta-product identities. We wish to prove an identity of the form

$$(3.9) \quad U_5(g) = \alpha_1 f_1(\tau) + \alpha_2 f_2(\tau) + \dots + \alpha_n f_n(\tau),$$

where  $p$  is prime,  $p|N$ ,  $g(\tau)$  is an eta-product and a modular function on  $\Gamma_0(pN)$ , and each  $f_j(\tau)$  is an eta-product and a modular function on  $\Gamma_0(N)$ . We use the following algorithm:

STEP 0. Write the identity in the form (3.9).

STEP 1. Use Theorem 2.1 to check that  $f_j(\tau)$  is a modular function on  $\Gamma_0(N)$  for each  $1 \leq j \leq n$ , and  $g(\tau)$  is a modular function on  $\Gamma_0(pN)$ .

STEP 2. Use Theorem 2.2 to find a set  $S_N$  of inequivalent cusps for  $\Gamma_0(N)$  and the fan width of each cusp.

STEP 3a. We compute  $ORD(f_j, \zeta, \Gamma_0(N))$  for each  $j$  and each cusp  $\zeta$  of  $\Gamma_0(N)$  apart from  $\infty$ .

STEP 3b. Use Theorem 3.4 to find a lower bound for

$$ORD(U_p g, r, \Gamma_0(N))$$

for each cusp  $r$  of  $\Gamma_0(N)$ . Call this lower bound  $L(g, r, N)$ .

STEP 4. Calculate

$$B = \sum_{s \in S_N, s \neq \infty} \min ( \{ ORD(f_j, s, \Gamma_0(N)) : 1 \leq j \leq n \} \cup \{ L(g, s, N) \} ).$$

STEP 5. Show that  $ORD(h(\tau), \infty, \Gamma_0(N)) > -B$ ,

where  $h(\tau) = U_p(g(\tau)) - (\alpha_1 f_1(\tau) + \alpha_2 f_2(\tau) + \dots + \alpha_n f_n(\tau))$ .

Corollary 3.3 then implies that  $h(\tau) \equiv 0$  and hence the eta-product identity (3.9).

EXAMPLE:

Let

$$g(\tau) = \frac{\eta(50\tau)^5 \eta(5\tau)^4 \eta(4\tau)^3 \eta(2\tau)^3}{\eta(100\tau)^3 \eta(25\tau)^2 \eta(10\tau)^8 \eta(\tau)^2}.$$

We prove that

$$U_5(g(\tau)) = 5f_1(\tau) + 2f_2(\tau),$$

where

$$f_1(\tau) = \frac{\eta(10\tau)^8 \eta(\tau)^4}{\eta(5\tau)^4 \eta(2\tau)^8}, \quad f_2(\tau) = \frac{\eta(10\tau)^5 \eta(\tau)^2}{\eta(20\tau)^3 \eta(5\tau)^2 \eta(4\tau) \eta(2\tau)}.$$

STEP 1. Use Theorem 2.1 to check that  $f_j(\tau)$  is a modular function on  $\Gamma_0(20)$  for each

$1 \leq j \leq 2$ , and  $g(\tau)$  is a modular function on  $\Gamma_0(100)$ .

```
> gpg:=[100, -3, 50, 5, 25, -2, 10, -8, 5, 4, 4, 3, 2, 3, 1, -2]:
```

```
> epg:=gp2etaproduct(gpg);
```

$$epg := \frac{\eta(50\tau)^5 \eta(5\tau)^4 \eta(4\tau)^3 \eta(2\tau)^3}{\eta(100\tau)^3 \eta(25\tau)^2 \eta(10\tau)^8 \eta(\tau)^2} \quad (50)$$

```
> gammacheck(gpg,100);
```

1

(51)

```
> gpf1:=[10, 8, 5, -4, 2, -8, 1, 4]:epf1:=gp2etaproduct(gpf1);
```

$$epf1 := \frac{\eta(10\tau)^8 \eta(\tau)^4}{\eta(5\tau)^4 \eta(2\tau)^8} \quad (52)$$

```
> gpf2:=[20, -3, 10, 5, 5, -2, 4, -1, 2, -1, 1, 2]:
```

```
> epf2:=gp2etaproduct(gpf2);
```

$$epf2 := \frac{\eta(10\tau)^5 \eta(\tau)^2}{\eta(20\tau)^3 \eta(5\tau)^2 \eta(4\tau) \eta(2\tau)} \quad (53)$$

```
> gammacheck(gpf1,20),gammacheck(gpf2,20);
```

1, 1

(54)

STEP 2. Use Theorem 2.2 to find a set  $S_{20}$  of inequivalent cusps for  $\Gamma_0(20)$  and the fan width of each cusp.

```
> C20:=cuspmake(20);
```

$$C20 := \left\{ 0, \frac{1}{2}, \frac{1}{4}, \frac{1}{5}, \frac{1}{10}, \frac{1}{20} \right\} \quad (55)$$

```
> seq([cusp,fanwidth(cusp,20)], cusp in C20);
```

$$[0, 20], \left[ \frac{1}{2}, 5 \right], \left[ \frac{1}{4}, 5 \right], \left[ \frac{1}{5}, 4 \right], \left[ \frac{1}{10}, 1 \right], \left[ \frac{1}{20}, 1 \right] \quad (56)$$

STEP 3a. We compute  $ORD(f_j, \zeta, \Gamma_0(20))$  for each  $j$  and each cusp  $\zeta$  of  $\Gamma_0(20)$  apart from  $\infty$ .

```
> C20:=cuspmake(20) minus {1/20}:
```

```
> ORDS1:=cuspORDSnotoo(epf1,C20,20);
```

$$ORDS1 := \left[ [0, 0], \left[ \frac{1}{2}, -2 \right], \left[ \frac{1}{4}, -2 \right], \left[ \frac{1}{5}, 0 \right], \left[ \frac{1}{10}, 2 \right] \right] \quad (57)$$

```
> ORDS2:=cuspORDSnotoo(epf2,C20,20);
```

$$ORDS2 := \left[ [0, 1], \left[ \frac{1}{2}, 0 \right], \left[ \frac{1}{4}, -1 \right], \left[ \frac{1}{5}, 0 \right], \left[ \frac{1}{10}, 1 \right] \right] \quad (58)$$

STEP 3b. Use Theorem 3.4 to find a lower bound for

$$ORD(U_p g, r, \Gamma_0(20))$$

for each cusp  $r$  of  $\Gamma_0(20)$ .

```
> C20:=cuspmake(20) minus {1/20}:
```

```
> ORDSg:=[seq([cusp,UpLB(epg,cusp,100,5)], cusp in C20)];
```

$$ORDSg := \left[ [0, 0], \left[ \frac{1}{2}, -2 \right], \left[ \frac{1}{4}, -2 \right], \left[ \frac{1}{5}, -\frac{1}{5} \right], \left[ \frac{1}{10}, \frac{3}{5} \right] \right] \quad (59)$$

STEP 4. Calculate the constant  $B$  in (3.10).

```
> mintotGAMMA0ORDS([ORDSg,ORDS1,ORDS2], 3);
```

$$-\frac{18}{5} \quad (60)$$

STEP 5. To prove the identity (3.11) we need to verify that

$$ORD(h(\tau), \infty, \Gamma_0(20)) > 3,$$

where

$$h(\tau) = U_5(g(\tau)) - (5f_1(\tau) + 2f_2(\tau)).$$

```
> U5g:=sift(etaproductoqseries(epg,1010),q,5,0,1000):
```

```
> h:=U5g - qetacombo(5*epf1 + 2*epf2,210):
```

```
> series(h,q,201);
```

$$O(q^{201}) \quad (61)$$

This completes the proof of the identity (3.11). We only had to show that the coefficient of  $q^j$  was zero in the  $q$ -expansion of  $h(\tau)$  for  $j \leq 4$ . We actually did it for  $j \leq 200$  as a check.

`provemodfuncGAMMA0UpETAid(EP,p,etacombo,N)` --- attempts to prove the identity

$$U_p(EP) = etacombo,$$

where  $EP$  is an eta-product and a modular function on  $\Gamma_0(pN)$ ,  $p$  is prime,  $p|N$ , and `etacombo` is a linear combination of eta-products which are all modular functions on  $\Gamma_0(N)$ . It returns the constant  $B$  in equation (3.10) and prints (if possible) the details of the verification and proof of the identity. If `xprint=true` then more details of the verification are printed. When this function is called there is a query asking whether to verify the identity.

Enter `yes` to carry out the verification.

```
> etacombo:=5*epf1 + 2*epf2:
```

```
> provemodfuncGAMMA0UpETAid(epg,5,etacombo,20);
```

```

*** There were NO errors.
*** o EP is an MF on Gamma[0](100)
*** o Each term in the etacombo is a modular function on
    Gamma0(20).
*** o We also checked that the total order of
    each term etacombo was zero.
*** To prove the identity U[5](EP)=etacombo we need to show
    that v[oo](ID) > 3 This means checking up to q^(4).
Do you want to prove the identity? (yes/no)
You entered yes.
We verify the identity to O(q^(43)).
We find that LHS - RHS is

```

$$O(q^{43})$$

RESULT: The identity holds to  $O(q^{43})$ .

CONCLUSION: This proves the identity since we had only  
to show that  $v[oo](ID) > 3$ .

**provemodfuncGAMMA0UpETAidBATCH(EP,p,etacombo,N)** --- is a version of  
**provemodfuncGAMMA0UpETAid** that prints less detail and does not query.

```
> noprint:=false:
```

```
> provemodfuncGAMMA0UpETAidBATCH(epg,5,etacombo,20);
```

```

*** There were NO errors.
*** o EP is an MF on Gamma[0](100)
*** o Each term in the etacombo is a modular function on
    Gamma0(20).
*** o We also checked that the total order of
    each term etacombo was zero.
*** To prove the identity U[5](EP)=etacombo we need to show
    that v[oo](ID) > 3 This means checking up to q^(4).
We find that LHS - RHS is

```

$$O(q^{43})$$

$$[1, -3, O(q^{43})]$$

(62)

Let  $L = [1, -3, O(q^{43})]$ .  $L[1] = 1$  means the identity is proved.  $L[2] = B$  (the constant in equation (3.2)), and we see that  $B = -2$ . This confirms an earlier calculation using **mintotORDS**.  $L[3] = O(q^{43})$  means the identity was checked up to  $q^{42}$ . To print out minimal information set **noprint:=true**.

```
> noprint:=true:
```

```
> provemodfuncGAMMA0UpETAidBATCH(epg,5,etacombo,20);
```

$$[1, -3, O(q^{43})]$$

(63)

#### REFERENCES

[1] Anthony J. F. Biagioli, *A proof of some identities of Ramanujan using modular forms*, Glasgow Math. J. **31** (1989), no. 3, 271-295. [MR1021804](#)

[2] Kok Seng Chua and Mong Lung Lang, *Congruence subgroups associated to the monster*, Experiment. Math. **13** (2004), no. 3, 343-360. [MR2103332](#)

[3] Frank Garvan, *A q-product tutorial for a q-series MAPLE package*, Sem. Lothar. Combin. **42** (1999), Art. B42d, 27 pp. (electronic), The Andrews Festschrift (Maratea, 1998). [MR1701583](#)

[4] B. Gordon and K. Hughes, *Ramanujan congruences for q(n)*, Analytic number theory (Philadelphia, Pa., 1980), Lecture Notes in Math., vol. 899, Springer, Berlin-New York, 1981, pp. 333-359.

[5] Frank Garvan, *A q-product tutorial for a q-series MAPLE package*, Sem. Lothar. Combin. **42** (1999), Art. B42d, 27 pp. (electronic), The Andrews Festschrift (Maratea, 1998). [MR1701583](#)

[6] B. Gordon and K. Hughes, *Ramanujan congruences for q(n)*, Analytic number theory (Philadelphia, Pa., 1980), Lecture Notes in Math., vol. 899, Springer, Berlin-New York, 1981, pp. 333-359.



[MR654539](#)

[5] Gerard Ligozat, *Courbes modulaires de genre 1*, Societe Mathematique de France, Paris, 1975, Bull. Soc. Math. France, Mem. 43, Supplement au Bull. Soc. Math. France Tome 103, no. 3. [MR0417060](#)

[6] Morris Newman, *Construction and application of a class of modular functions. II*, Proc. London Math. Soc. (3) **9** (1959), 373-387. [MR0107629](#)

[7] Ken Ono, *The web of modularity: arithmetic of the coefficients of modular forms and q-series*, CBMS Regional Conference Series in Mathematics, vol. 102, Published for the Conference Board of the Mathematical Sciences, Washington, DC; by the American Mathematical Society, Providence, RI, 2004.

[MR2020489](#)

[8] Robert A. Rankin, *Modular forms and functions*, Cambridge University Press, Cambridge, 1977.

[MR0498390](#)